3

Electrical Characteristics of the Solar Cell

Summary

The basic equations of a solar cell are described in this chapter. The dark and illuminated $I(V)$ characteristics are analytically described and PSpice models are introduced firstly for the simplest model, composed of a diode and a current source. The fundamental electrical parameters of the solar cell are defined: short circuit current (I_{sc}), open circuit voltage (V_{oc}), maximum power (P_{max}) and fill factor (FF). This simple model is then generalized to take into account series and shunt resistive losses and recombination losses. Temperature effects are then introduced and the effects of space radiation are also studied with a modification of the PSpice model. A behavioural model is introduced which allows the solar cell simulation for arbitrary time profiles of irradiance and temperature.

3.1 Ideal Equivalent Circuit

As shown in Chapter 2, a solar cell can in a first-order model, be described by the superposition of the responses of the device to two excitations: voltage and light. We start by reproducing here the simplified equation governing the current of the solar cell, that is equation (2.16)

$$J = J_{sc} - J_0(e^{\frac{V}{V_T}} - 1) \tag{3.1}$$

which gives the current density of a solar cell submitted to a given irradiance and voltage. The value of the current generated by the solar cell is given by

$$I = I_{sc} - I_0(e^{\frac{V}{V_T}} - 1) \tag{3.2}$$

where I_{sc} and I_0 relate to their respective current densities J_{sc} and J_0 as follows:

$$I_{sc} = AJ_{sc} \tag{3.3}$$

$$I_0 = AJ_0 \tag{3.4}$$

where A is the total area of the device. The metal covered area has been neglected.

As can be seen, both the short circuit current and the dark current scale linearly with the solar cell area, and this is an important result which facilitates the scaling-up or down of PV systems according to the requirements of the application.

This is the simplest, yet the most used model of a solar cell in photovoltaics and its applications, and can be easily modelled in PSpice code by a current source of value I_{sc} and a diode.

Although I_0 is a strong function of the temperature, we will consider first that I_0 can be given a constant value. Temperature effects are addressed later in this chapter.

3.2 PSpice Model of the Ideal Solar Cell

As has already been described, one way to handle a PSpice circuit is to define subcircuits for the main blocks. This is also the case of a solar cell, where a subcircuit facilitates the task of connecting several solar cells in series or in parallel as will be shown later.

The PSpice model of the subcircuit of an ideal solar cell is shown in Figure 3.1(a) which is the circuit representation of equation (3.2). The case in photovoltaics is that a solar cell receives a given irradiance value and that the short circuit current is proportional to the irradiance. In order to implement that in PSpice the value of the short circuit current, is assigned to a G-device which is a voltage-controlled current source, having a similar syntax to the e-devices:



Figure 3.1 (a) Cell_1.lib subcircuit a solar cell and (b) block diagram

Syntax for G-device

```
g_name node+ node− control_node+ control_node− gain
```

As can be seen, this is a current source connected to the circuit between nodes 'node+' and 'node−', with a value given by the product of the gain by the voltage applied between control_node+ and control_node−.

A simplification of this device consists of assigning a value which can be a mathematical expression as follows:

```
g_name node+ node− value = {expression}
```

In our case the G-device used is named 'girrad' and is given by:

$$girrad = \frac{J_{sc}A}{1000}G \tag{3.5}$$

where G is the value of the irradiance in W/m^2. Equation (3.5) considers that the value of J_{sc} is given at standard (AM1.5G, 1000 W/m^2 $T_{cell} = 25\,°C$) conditions, which are the conditions under which measurements are usually made. Solar cell manufacturer's catalogues provide these standard values for the short circuit current. Equation (3.5) returns the value of the short circuit current at any irradiance value G, provided the proportionality between irradiance and short circuit current holds. This is usually the case provided low injection conditions are satisfied.

The subcircuit netlist follows:

```
.subckt cell_1 300 301 302 params: area=1, j0=1, jsc=1

girrad 300 301 value={(jsc/1000)*v(302)*area}
d1 301 300 diode
.model diode d(is={j0*area})
.ends cell_1
```

As can be seen, dummy values, namely unity, are assigned to the parameters at the subcircuit definition; the real values are specified later when the subcircuit is included in circuit. The diode model includes the definition of the parameter 'is' of a PSpice diode as the result of the product of the saturation current density J_0 multiplied by the area A. The block diagram is shown in Figure 3.1(b). As can be seen there is a reference node (300), an input node (302) to input a voltage numerically equal to the irradiance value, and an output node (301) which connects the solar cell to the circuit.

The solar cell subcircuit is connected in a measurement circuit in order to obtain the $I(V)$ characteristic. This is accomplished by the circuit shown in Figure 3.2, where the solar cell area corresponds to a 5″ diameter device. A short circuit density current of 34.3 mA/cm^2 ha



Figure 3.2 Measurement circuit of the $I(V)$ characteristics of a solar cell

been assumed and $J_0 = 1 \times 10^{-11}$ A/cm^2 is chosen as an example leading to a PSpice file as follows:

```
*cell_1.cir
.include cell_1.lib
xcell1 0 31 32 cell_1 params:area=126.6  j0=1e-11  jsc=0.0343
vbias 31 0 dc 0
virrad 32 0 dc 1000

.plot dc i(vbias)
.probe
.dc vbias -0.1 0.6 0.01
.end
```

As can be seen the circuit includes a DC voltage source 'vbias' which is swept from −0.5 V to + 0.6 V. The result of the simulation of the above netlist is shown in Figure 3.3. The intersection of the graph with the y-axis provides the value of the short circuit current of the cell, namely 4.342 A, which is, of course, the result of $0.0343 \times 126.6 = 4.342$ A. Some other important points can be derived from this $I(V)$ plot as described in the next section.



Figure 3.3 $I(V)$ characteristics of the solar cell model in Figure 3.2

3.3 Open Circuit Voltage

Besides the short circuit current, a second important point in the solar cell characteristics can be defined at the crossing of the $I(V)$ curve with the voltage axis. This is called the open circuit point and the value of the voltage is called the open circuit voltage, V_{oc}.

Applying the open circuit condition, $I = 0$, to the $I(V)$ equation (3.2) as follows:

$$I = 0 = I_{sc} - I_0\left(e^{\frac{V_{oc}}{V_T}} - 1\right) \tag{3.6}$$

the open circuit voltage is given by:

$$V_{oc} = V_T \ln\left(1 + \frac{I_{sc}}{I_0}\right) \tag{3.7}$$

From equation (3.7), it can be seen than the value of the open circuit voltage depends, logarithmically on the I_{sc}/I_0 ratio. This means that under constant temperature the value of the open circuit voltage scales logarithmically with the short circuit current which, in turn scales linearly with the irradiance resulting in a logarithmic dependence of the open circuit voltage with the irradiance. This is also an important result indicating that the effect of the irradiance is much larger in the short circuit current than in the open circuit value.

Substituting equations (3.3) and (3.4) in equation (3.7), results in:

$$V_{oc} = V_T \ln\left(1 + \frac{J_{sc}}{J_0}\right) \tag{3.8}$$

The result shown in equation (3.8) indicates that the open circuit voltage is *independent* of the cell area, which is an important result because, regardless of the value of the cell area, the open circuit voltage is always the same under the same illumination and temperature conditions.

Example 3.1

Consider a circular solar cell of 6″ diameter. Assuming that $J_{sc} = 34.3$ A/cm^2 and $J_0 = 1 \times 10^{-11}$ A/cm^2, plot the $I(V)$ characteristic and calculate the open circuit voltage for several irradiance values, namely $G = 200, 400, 600, 800$ and 1000 W/m^2.

The netlist in this case has to include a instruction to solve the circuit for every value of the irradiance. This is performed by the instructions below:



igure 3.4 $I(V)$ plots of a solar cell under several irradiance values: 200, 400, 600, 800 and 000 W/m^2

```
*IRRADIANCE.CIR
.include cell_1.lib
xcell1 0 31 32 cell_1 params:area=126.6 j0=1e-11 jsc=0.0343
vbias 31 0 dc 0
.param IR=1
virrad 32 0 dc {IR}
.step param IR list 200 400 600 800 1000
.plot dc i(vbias)
.probe
.dc vbias -0.1 0.6 0.01
.end
```

It can be seen that a new parameter is defined, IR, which is assigned to the value of the ltage source virrad. The statement '.step param' is a dot command which makes PSpice peat the simulation for all values of the list.

The plots in Figure 3.4 are obtained and, using the cursor utility in Probe, the values of V_{oc} d I_{sc} are measured. The results are shown in Table 3.1.

Table 3.1 Short circuit current and open circuit voltages for several irradiance values

idiance W/m^2)	Short circuit current I_{sc}(A)	Open circuit voltage I_{oc}(V)
000	4.34	0.567
300	3.47	0.561
500	2.60	0.554
400	1.73	0.543
200	0.86	0.525

3.4 Maximum Power Point

The output power of a solar cell is the product of the output current delivered to the electric load and the voltage across the cell. It is generally considered that a positive sign indicates power being delivered to the load and a negative sign indicates power being consumed by the solar cell. Taking into account the sign definitions in Figure 3.1, the power at any point of the characteristic is given by:

$$P = V \times I = V\left[I_L - I_0\left(e^{\frac{V}{V_T}} - 1\right)\right] \tag{3.9}$$

Of course, the value of the power at the short circuit point is zero, because the voltage is zero, and also the power is zero at the open circuit point where the current is zero. There is a positive power generated by the solar cell between these two points. It also happens that there is a maximum of the power generated by a solar cell somewhere in between. This happens at a point called the maximum power point (MPP) with the coordinates $V = V_m$ and $I = I_m$. A relationship between V_m and I_m can be derived, taking into account that at the maximum power point the derivative of the power is zero:

$$\frac{dP}{dV} = 0 = I_L - I_0\left(e^{\frac{V_m}{V_T}} - 1\right) - \frac{V_m}{V_T} I_0 e^{\frac{V_m}{V_T}} \tag{3.10}$$

at the MPP,

$$I_m = I_L - I_0\left(e^{\frac{V_m}{V_T}} - 1\right) \tag{3.11}$$

it follows that,

$$V_m = V_{oc} - V_T \ln\left(1 + \frac{V_m}{V_T}\right) \tag{3.12}$$

which is a transcendent equation. Solving equation (3.13) V_m can be calculated provided V_{oc} is known.

Using PSpice the coordinates of the MMP can be easily found plotting the $I \times V$ product as a function of the applied voltage. We do not need to write a new PSpice '.cir' file but just draw the plot. This is shown in Figure 3.5 for the solution of Example 3.1.

Table 3.2 shows the values obtained using equation (3.12) compared to the values obtained using PSpice. As can be seen the accuracy obtained by PSpice is related to the step in voltage we have used, namely 0.01 V. If more precision is required, a shorter simulation step should be used.

Other models can be found in the literature [3.11] to calculate the voltage V_m, which are summarized below.



re 3.5 Plot of the product of the current by the voltage across the solar cell for several values of irradiance, namely: 200, 400, 600, 800 and 1000 W/m^2

lternative model number 1:

$$V_m = V_{oc} - 3V_T \tag{3.13}$$

lternative model number 2:

$$\frac{V_m}{V_{oc}} = 1 - \left(\frac{1+\ln\beta}{2+\ln\beta}\right)\frac{\ln(1+\ln\beta)}{\ln\beta} \tag{3.14}$$

h

$$\beta = \frac{I_{sc}}{I_0} \tag{3.15}$$

Table 3.2 PSpice results for several irradiance values

diance /m^2)	V_m (PSpice) (V)	V_m from equation (3.14) (V)	I_m (PSpice) (A)	P_{max} (PSpice) (W)
000	0.495	0.4895	4.07	2.01
800	0.485	0.4825	3.28	1.59
600	0.477	0.476	2.47	1.18
400	0.471	0.466	1.63	0.769
200	0.45	0.4485	0.820	0.37

3.5 Fill Factor (FF) and Power Conversion Efficiency (η)

A parameter called fill factor (FF) is defined as the ratio between the maximum power P_{max} and the $I_{sc}V_{oc}$ product:

$$FF = \frac{V_m I_m}{V_{oc} I_{sc}} \tag{3.16}$$

The fill factor has no units, indicating how far the product $I_{sc}V_{oc}$ is from the power delivered by the solar cell.

The FF can also be approximated by an empirical relationship as follows [3.2],

$$FF_0 = \frac{v_{oc} - \ln(v_{oc} + 0.72)}{1 + v_{oc}} \tag{3.17}$$

Sub-index 0 indicates that this is the value of the FF for the ideal solar cell without resistive effects to distinguish it from the FF of a solar cell with arbitrary values of the losses resistances, which will be addressed in the sections below. The parameter v_{oc} is the normalized value of the open circuit voltage to the thermal potential V_T, as

$$v_{oc} = \frac{V_{oc}}{V_T} \tag{3.18}$$

Equation (3.17) gives reasonable accuracy for v_{oc} values greater than 10.

Example 3.2

From the PSpice simulations in Example 3.1 obtain the values of the FF for several values of the irradiance.

Taking the data for I_{sc} and V_{oc} from Table 3.1 and the data for V_m and I_m from Table 3.2 the values for FF are easily calculated and are shown in Table 3.3.

As can be seen, the FF is reasonably constant for a wide range of values of the irradiance and close to 0.8.

Table 3.3 Fill factor values for Example 3.1

Irradiance (W/m^2)	Fill factor
1000	0.816
800	0.816
600	0.819
400	0.818
200	0.819

Table 3.4 Fill factor values for Example 3.1

Irradiance (W/m^2)	Power conversion efficiency (%)
1000	15.8
800	15.68
600	15.53
400	15.17
200	14.6

The power conversion efficiency η is defined as the ratio between the solar cell output ver and the solar power impinging the solar cell surface, P_{in}. This input power equals the diance multiplied by the cell area:

$$\eta = \frac{V_m I_m}{P_{in}} = FF\frac{V_{oc}I_{sc}}{P_{in}} = FF\frac{V_{oc}I_{sc}}{G \times Area} = FF\frac{V_{oc}J_{sc}}{G} \tag{3.19}$$

As can be seen the power conversion efficiency of a solar cell is proportional to the value the three main photovoltaic parameters: short circuit current density, open circuit voltage l fill factor, for a given irradiance G.

Most of the time the efficiency is given in %.

In Example 3.2 above, the values of the efficiency calculated from the results in Tables 3.2 l 3.3 are given in Table 3.4.

As can be seen the power conversion efficiency is higher at higher irradiances. This can be lytically formulated by considering that the value of the irradiance is scaled by a 'scale tor S' to the standard conditions: 1 Sun, AM1.5 1000 W/m^2, as follows:

$$G = SG_1 \tag{3.20}$$

h $G_1 = 1000$ W/m^2. The efficiency is now given by:

$$\eta = FF\frac{V_{oc}I_{sc}}{SG_1 \times Area} \tag{3.21}$$

f proportionality between irradiance and short circuit current can be assumed:

$$I_{sc} = SI_{sc1}$$
$$V_{oc} = V_T \ln\left(1 + \frac{SI_{sc1}}{I_0}\right) \approx V_T \ln\left(\frac{SI_{sc1}}{I_0}\right) = V_T \ln\left(\frac{I_{sc1}}{I_0}\right) + V_T \ln S \tag{3.22}$$

l finally

$$\eta = FF\frac{SI_{sc1}}{SG_1 \times Area}\left[V_T \ln\left(\frac{I_{sc1}}{I_0}\right) + V_T \ln S\right] = FF\frac{I_{sc1}V_{oc1}}{G_1 \times Area}\left[1 + \frac{V_T \ln S}{V_{oc1}}\right] \tag{3.23}$$

Applying the efficiency definition in equation (3.21) to the one-sun conditions

$$\eta_1 = FF_1\frac{V_{oc1}I_{sc1}}{G_1 \times Area} \tag{3.24}$$

it follows

$$\eta = \eta_1\left[1 + \frac{V_T \ln S}{V_{oc1}}\right] \tag{3.25}$$

provided that $FF = FF_1$, thereby indicating that the efficiency depends logarithmically on the value of the scale factor S of the irradiance. The assumptions made to derive equation (3.25), namely constant temperature is maintained, that the short circuit current is proportional to the irradiance and that the fill factor is independent of the irradiance value, have to be fulfilled. This is usually the case for irradiance values smaller than one sun.

3.6 Generalized Model of a Solar Cell

The equivalent circuit and the PSpice model of the solar cell described so far takes into account an ideal behaviour of a solar cell based on an ideal diode and an ideal current source. Sometimes this level-one model is insufficient to accurately represent the maximum power delivered by the solar cell. There are several effects which have not been taken into account and that may affect the solar cell response.

(a) Series resistance

One of the main limitations of the model comes from the series resistive losses which are present in practical solar cells. In fact, the current generated in the solar cell volume travels to the contacts through resistive semiconductor material, both, in the base region, not heavily doped in general, and in the emitter region, which although heavily doped, is narrow. Besides these two components, the resistance of the metal grid, contacts and current collecting bus also contribute to the total series resistive losses. It is common practice to assume that these series losses can be represented by a lumped resistor, R_s, called the series resistance of the solar cell.

(b) Shunt resistance

Solar cell technology in industry is the result of mass production of devices generally made out of large area wafers, or of large area thin film material. A number of shunt resistive losses are identified, such as localized shorts at the emitter layer or perimeter shunts along cell borders are among the most common. This is represented generally by a lumped resistor, R_{sh}, in parallel with the intrinsic device.

(c) Recombination

Recombination at the space charge region of solar cells explains non–ohmic current paths in parallel with the intrinsic solar cell. This is relevant at low voltage bias and can be represented in an equivalent circuit by a second diode term with a saturation density current J_{02}, which is different from the saturation density current of the ideal solar cell diode, and a given ideality diode factor different to 1, it is most often assumed to equal 2. This can be added to the solar cell subcircuit by simply adding a second diode 'diode2' with $n = 2$ in the description of the diode model, as follows:

```
.model diode2 d(is={j02*area}, n=2)
```

(d) Non-ideality of the diffusion diode

In practice few devices exhibit a totally ideal $I(V)$ characteristic with ideality coefficient equal to unity. For this reason it is common practice to also add a parameter 'n' to account for hese non–idealities. In the same way as described above for the recombination diode, the main diode model can be modified to take into account this effect as follows:

```
.model diode d(is={j0*area},n=1.1)
```

where $n = 1.1$ is an example.

In summary, a new relationship between current and voltage can be written taking into account these effects as follows:

$$I = I_L - I_0\left(e^{\frac{V+IR_s}{nV_T}} - 1\right) - I_{02}\left(e^{\frac{V+IR_s}{2V_T}} - 1\right) - \frac{V + IR_s}{R_{sh}} \tag{3.26}$$

This equation comes from the equivalent circuit shown in Figure 3.6. It must be emphasized that the meaning of the short circuit current has to change in this new circuit due to the fact that the short circuit conditions are applied to the external solar cell terminals, namely nodes (303) and (300) in Figure 3.6, whereas equations (2.2), (2.3) and (2.6) in Chapter 2 were derived for a short circuit applied to the internal solar cell nodes, namely (301) and (300) in Figure 3.6. This is the reason why the current generator is now named I_L, or photogenerated current, to differentiate it from the new short circuit current of the solar cell. Finally, the diode D1 implements the first diode term in equation (3.26), and the diode D2 implements the second diode. It will be shown in the examples below how the values of the several parameters involved in the model influence the solar cell response, in particular the differences between photogenerated current and short circuit current.



Figure 3.6 Subcircuit cell_2.lib including two diodes and series and shunt resistors

3.7 Generalized PSpice Model of a Solar Cell

The circuit in Figure 3.6, which includes the new subcircuit and the series and resistors, can be described in PSpice code by the following netlist:

```
*CELL_2.LIB*
.subckt cell_2 300 303 302 params:area=1, j0=1, jsc=1, j02=1, rs=1, rsh=1

girrad 300 301 value={(jsc/1000)*v(302)*area}
d1 301 300 diode
.model diode d(is={j0*area})
d2 11 10 diode2
.model diode2 d(is={j02*area}, n=2)
.ends cell_2
```

where new model parameters are j02, rs and rsh which are given appropriate values. In to plot the $I(V)$ curve and analyse the effects of the values of the new parameters a .c has to be written calling the new subcircuit. This is shown in the next section.

3.8 Effects of the Series Resistance on the Short-Circuit Current and the Open-Circuit Voltage

The simulation results are shown in Figure 3.7 for several values of the series resistance at a high and constant shunt resistance, namely $1\times10^5\,\Omega$, and at equivalent values of the irradiance and the temperature.

This is achieved by changing the statement calling the subcircuit, introducing {RS} as the value for the series resistance, which is then given several values at the parametric analysis statement '.step param', as follows,

```
*cell_2.cir
.include cell_2.lib
xcell2 0 31 32 cell_2 params:area=126.6  j0=1e-11  j02=1E-9
+ jsc=0.0343 rs={RS} rsh=100000
.param RS=1
vbias 31 0 dc 0
virrad 32 0 dc 1000
.plot dc i(vbias)
.dc vbias -0.1 0.6 0.01
.step param RS list 0.0001 0.001 0.01 0.1 1
.probe
.end
```



Figure 3.7 Series resistance effects. $I(V)$ characteristics for series resistance values from 1 Ω (bottom graph), 0.1 Ω, 0.01 Ω, 0.001 Ω, 0.0001 Ω (top graph)

As can be seen in Figure 3.7, large differences are observed in the $I(V)$ characteristics as the value of the series resistance increases, in particular the values of the short-circuit current and of the fill factor can be severely reduced. Some of these changes can be explained using equation (3.26) and are summarized and explained next.

Short-circuit current

In contrast with the ideal result, at short circuit the value of the short-circuit current is not equal to the photocurrent I_L. By replacing $V=0$ in equation (3.26) it follows that:

$$I_{sc} = I_L - I_0\left(e^{\frac{I_{sc}R_s}{nV_T}} - 1\right) - I_{02}\left(e^{\frac{I_{sc}R_s}{2V_T}} - 1\right) - \frac{I_{sc}R_s}{R_{sh}} \tag{3.27}$$

In practice, the difference between I_{SC} and I_L in a real solar cell is small because the series resistance is kept low by a proper design of the metal grid and doping levels, and because the parallel resistance is large in devices passing the post-fabrication tests. This is clearly seen in the several plots in Figure 3.7 where the value of the short-circuit current remains sensibly constant, provided that values for the series resistance equal to or smaller than 0.01 Ω are used. Taking into account that the solar cell has an area, in this example, of 126.6 cm^2, this means that very small values of the series resistive losses can be tolerated if the short-circuit current is to be maintained.

Open-circuit voltage

The open-circuit voltage can be derived from equation (3.26) setting $I=0$.

$$0 = I_L - I_0\left(e^{\frac{V_{oc}}{nV_T}} - 1\right) - I_{02}\left(e^{\frac{V_{oc}}{2V_T}} - 1\right) - \frac{V_{oc}}{R_{sh}} \tag{3.28}$$

It is clear that the open-circuit voltage given by equation (3.28) is independent of the series resistance value and this is true regardless of the value of the shunt resistance, and the values of the parameters of the recombination diode. If we neglect the third and fourth terms in equation (3.28), the open-circuit voltage is given by

$$V_{oc} = nV_T \ln\left(1 + \frac{I_{sc}}{I_0}\right) \tag{3.29}$$

which is sensibly the same result found using the ideal equivalent circuit in equation (3.7) with the exception of the non-ideality factor of the main diode, n.

In Figure 3.7 the graphs shown confirm that the open-circuit voltage is independent of the series resistance value, as all curves cross at the same point on the voltage axis.

3.9 Effect of the Series Resistance on the Fill Factor

From the results shown in Figure 3.7 it becomes clear that one of the most affected electrical parameters of the solar cell by the series resistance is the fill factor.

If we restrict the analysis to the effect of the series resistance only, the solar cell can be nodelled by the diffusion diode and the series resistance, simplifying equation (3.26), which hen becomes,

$$I = I_L - I_0\left(e^{\frac{V+IR_s}{nV_T}} - 1\right) \tag{3.30}$$

n open circuit conditions:

$$0 = I_L - I_0\left(e^{\frac{V_{oc}}{nV_T}} - 1\right) \tag{3.31}$$

rom equation (3.31), I_0 can be written as

$$I_0 = \frac{I_L}{\left(e^{\frac{V_{oc}}{nV_T}} - 1\right)} \approx I_L e^{-\frac{V_{oc}}{nV_T}} \tag{3.32}$$

nd substituting in equation (3.30)

$$I = I_L - I_L e^{-\frac{V_{oc}}{nV_T}}\left(e^{\frac{V+IR_s}{nV_T}} - 1\right) \approx I_L\left(1 - e^{\frac{V+IR_s-V_{oc}}{nV_T}}\right) \tag{3.33}$$

the maximum power point

$$I_m \approx I_L\left(1 - e^{\frac{V_m+IR_s-V_{oc}}{nV_T}}\right) \tag{3.34}$$

ultiplying equation (3.34) by the voltage V and making the derivative of the product equal zero, the maximum power point coordinates are related by

$$I_m + (I_m - I_L)\left(\frac{V_m - I_m R_s}{nV_T}\right) = 0 \tag{3.35}$$

V_{oc} and I_L are known, equations (3.35) and (3.34) give the values of I_m and V_m.

A simplified formulation of the same problem assumes that the maximum power delivered the solar cell can be calculated from

$$P'_m = P_m - I_m^2 R_s \tag{3.36}$$

here P'_m is the maximum power when R_s is not zero. This equation implies that the aximum power point shifts approximately at the same current value I_m.

Multiplying and dividing the second term of the right-hand side in equation (3.36) by V_m, it follows that

$$P'_m = P_m\left(1 - \frac{I_m}{V_m}R_s\right) \tag{3.37}$$

If we further assume that

$$\frac{I_m}{V_m} \approx \frac{I_{sc}}{V_{oc}} \tag{3.38}$$

then equation (3.38) becomes

$$P'_m = P_m\left(1 - \frac{I_{sc}}{V_{oc}}R_s\right) = P_m(1 - r_s) \tag{3.39}$$

where

$$r_s = \frac{R_s}{V_{oc}/I_{sc}} \tag{3.40}$$

is the normalized value of the series resistance. The normalization factor is the ratio of the open-circuit voltage to the short-circuit current.

The result in equation (3.40) is easily translated to the fill factor:

$$FF = \frac{P'_m}{V_{oc}I_{sc}} = \frac{P_m(1 - r_s)}{V_{oc}I_{sc}} = FF_o(1 - r_s) \tag{3.41}$$

relating the value of the FF for a non-zero value of the series resistance FF with the value of the FF when the series resistance is zero, FF_0. This result is valid provided that both the value of the short-circuit current and of the open-circuit voltage are independent of the value of the series resistance.

Example 3.3

Considering the solar cell data and PSpice model given above in this section, simulate a solar cell with the following parameter values: area = 126.6, $J_0 = 10^{-11}$, $J_{02} = 0$, $J_{sc} = 0.0343$, $R_{sh} = 100$, for the values of the series resistance, 10^{-4}, 10^{-3}, 2×10^{-3}, 5×10^{-3}, 10^{-2}, 2×10^{-2}, 5×10^{-2}, 10^{-1} Ω. Calculate the values of the FF and compare with the results given by equation (3.41).

We first start by writing the file shown in Annex 3 with name 'example3_3'. After running PSpice the values of the maximum power point are calculated and given in Table 3.5. A plot

Table 3.5 Results of Example 3.3

es resistance R_s	Maximum power P_m	FF	Normalized series resistance r_s	FF_o $(1-r_s)$
0.0001	2.02	0.82	7.69×10^{-4}	0.819
0.001	2.004	0.814	7.69×10^{-3}	0.813
0.002	1.98	0.8	0.0153	0.807
0.005	1.93	0.78	0.038	0.788
0.01	1.85	0.75	0.076	0.757
0.02	1.689	0.686	0.15	0.697



ure 3.8 Values of the fill factor plotted against the normalized value of the series resistance in ample 3.3. Solid line is calculated from values from equation (3.41) and dashed line is PSpice values

the FF value calculated after the PSpice simulation against the normalized value of the ies resistance is shown in Figure 3.8 and compared with the value given by equation 41). As can be seen for small values of the series resistance, the results agree quite well, d spread out as the series resistance increases.

.10 Effects of the Shunt Resistance

ie shunt resistance also degrades the performance of the solar cell. To isolate this effect om the others, the series resistance and the second diode can be eliminated from equation .26) by setting for R_s a very small value $R_s = 1\times10^{-6}\ \Omega$ and $J_{02} = 0$. The PSpice circuit n now be solved for several values of the shunt resistance as shown in Annex 3, file iunt.cir'.

The results are shown in Figure 3.9. It is clear that unless the parallel resistance takes very iall values, the open circuit voltage is only very slightly modified. On the other hand at



Figure 3.9 Effects of the shunt resistance on the $I(V)$ characteristics

short circuit conditions, all plots cross at the same point. This is not unusual because from equation (3.26), with $R_s = 0$ and $J_{02} = 0$, it results at $V = 0$:

$$I_{sc} = I_L \tag{3.42}$$

which is independent of R_{sh}. Additionally if we look at the $I(V)$ characteristic under reverse bias, $V \ll 0$

$$I = I_L - \frac{V}{R_{sh}} \tag{3.43}$$

indicating that the $I(V)$ characteristic should be a straight line with slope equal to $(-1/R_{sh})$. Small values of the shunt resistance also heavily degrade the fill factor.

3.11 Effects of the Recombination Diode

The open-circuit voltage is also degraded when the recombination diode becomes important. This can be seen in Figure 3.10 where in order to isolate the recombination diode effect, a high value of the parallel resistance and a low value of the series resistance have been selected.

Several values for the parameter J_{02} are plotted, namely 1×10^{-8} A/cm^2, 1×10^{-7} A/cm^2, 1×10^{-6} A/cm^2, 1×10^{-5} A/cm^2, 1×10^{-4} A/cm^2. The corresponding netlist is shown in Annex 3 called 'diode_rec.cir'.

The result indicates that when the recombination diode dominates, the characteristic is also heavily degraded, both in the open-circuit voltage and in the FF. The short-circuit current remains constant.



Figure 3.10 Effects of the recombination diode

12 Temperature Effects

erating temperature has a strong effect on the electrical response of solar cells. Taking ɔ account that in, terrestrial applications, solar cells can easily warm up to 60–65 °C and t in space or satellite applications temperatures can be even higher, it follows that a proper delling of the temperature coefficients of the main electrical parameters is mandatory.

Temperature effects in a solar cell can be included in the PSpice model by using the built- parameters of the diode model included in the equivalent circuit. Namely the saturation ısity current of a diode has a strong dependence on temperature and it is usually given by:

$$J_0 = BT^{XTI} e^{-\frac{E_g}{kT}} \tag{3.44}$$

ere B is a constant independent of the temperature and XTI is a PSpice parameter also ependent of the temperature. Equation (3.44) is valid for any arbitrary value of ıperature. It can also be written for a nominal or reference temperature T_{nom}, as follows

$$J_0(T_{nom}) = BT_{nom}^{XTI} e^{-\frac{E_g(T_{nom})}{kT_{nom}}} \tag{3.45}$$

Dividing equation (3.44) by equation (3.45):

$$\ln\left[\frac{J_0}{J_0(T_{nom})}\right] = XTI \ln\left(\frac{T}{T_{nom}}\right) + \frac{E_g(T_{nom})}{kT_{nom}} - \frac{E_g}{kT} \tag{3.46}$$

The value for the energy bandgap of the semiconductor E_g at any temperature T is given

$$E_g = E_{g0} - \frac{GAP1 \cdot T^2}{GAP2 + T} \tag{3.47}$$

ere E_{g0} is the value of the band gap extrapolated to $T = 0\,\mathrm{K}$ for the semiconductor ısidered.

Example 3.4

Plot the $I(V)$ characteristics of a silicon solar cell 5″ diameter, $J_{sc} = 0.343$ A/cm^2, $J_0 = 1 \times 10^{-11}$ assumed ideal for various temperatures, namely 27 °C, 35 °C, 40 °C, 45 °C, 50 °C, 55 °C and 60 °C. This analysis assumes that the short-circuit current is independent of temperature.

The netlist now has to be modified to include temperature analysis. This is highlighted in the netlist that follows:

```
*temp.cir
.include cell_3.lib
xcell13 0 31 32 cell_3 params:area=126.6 j0=1e-11 j02=0
+ jsc=0.0343 rs=1e-6 rsh=1000
vbias 31 0 dc 0
virrad 32 0 dc 1000
.plot dc i(vbias)
.dc vbias 0 0.6 0.01
.temp 27 35 40 45 50 55 60

.probe
.end
```

We have simplified the solar cell model setting $J_{02} = 0$, the series resistance has been set to a very small value and the shunt resistance to a large value so as not to obscure the temperature effects, which we highlight in this section. As can also be seen the subcircuit model has been changed by a 'cell_3.lib' file shown in Annex 3 to include, in the diode model, the silicon value for the parameter $E_{g0} = 1.17$ eV for silicon.

The results of the simulation are shown in Figure 3.11.

Using the cursor on the probe plots, the values of the open-circuit voltage for the various temperatures can be easily obtained. They are shown in Table 3.6.



Figure 3.11 Effects of temperature

Table 3.6 Values of the open-circuit voltage and temperature coefficient for several temperatures

Temperature (°C)	27	35	40	45	50	55	60
V_{oc}(mV)	567.5	550	538.4	526.9	515.3	503.8	492.9
$\Delta V_{oc}/\Delta T$ (mV/°C)		−2.18	−2.3	−2.3	−2.31	−2.31	−2.18

As can be seen the temperature coefficient of the open-circuit voltage, calculated in the second row in Table 3.6 is close to −2.3 mV/°C for a wide range of operating temperatures. This result is consistent with a theoretical deduction that can be made from the temperature derivative in equation (3.8) of the open-circuit voltage:

$$\frac{\mathrm{d}V_{oc}}{\mathrm{d}T} = \frac{V_T}{T}\ln\frac{J_{sc}}{J_0} + V_T\frac{\mathrm{d}}{\mathrm{d}T}\left(\ln\frac{J_{sc}}{J_0}\right) \tag{3.48}$$

Assuming that the short-circuit current is independent of the temperature, at least compared with the other contributions in equation (3.48), it follows:

$$\frac{\mathrm{d}V_{oc}}{\mathrm{d}T} = \frac{V_T}{T}\ln\frac{J_{sc}}{J_0} - V_T\frac{\mathrm{d}}{\mathrm{d}T}(\ln J_0) \tag{3.49}$$

Taking into account equation (3.44)

$$J_0 = CT^{\gamma}\mathrm{e}^{-\frac{E_{g0}}{kT}} \tag{3.50}$$

where C and γ are constants independent of the temperature. Replacing equation (3.50) in equation (3.49) and taking into account

$$V_{oc} = V_T\ln\left(1+\frac{J_{sc}}{J_0}\right) \approx V_T\ln\left(\frac{J_{sc}}{J_0}\right) \tag{3.51}$$

it follows,

$$\frac{\mathrm{d}V_{oc}}{\mathrm{d}T} = \frac{V_{oc}}{T} - \frac{\gamma V_T}{T} - \frac{E_{g0}}{qT} \tag{3.52}$$

Using typical silicon values for the parameters in equation (3.52), values around −2.3 mV/°C for $\gamma = 3$ are found for the temperature coefficient of the open-circuit voltage of silicon solar cells.

The assumption made about the independence of the short-circuit current on the temperature, although acceptable in many cases, can be removed by taking into account a

temperature coefficient of the short-circuit current. Typically for silicon solar cells a value of 6.4×10^{-6} A/cm^2°C is frequently used.

A modification to the subcircuit of the solar cell to account for this temperature coefficient of the short-circuit current can be made and it is shown in Example 3.5.

Example 3.5

(a) Write a PSpice subcircuit with two added parameters: temperature and short-circuit current temperature coefficient.

Solution

```
*cell_4.lib
.subckt cell_4 300 303 302 params:area=1, j0=1, jsc=1, j02=1, rs=1,
+rsh=1,temp=1,isc_coef=1

girrad 300 301 value={(jsc/1000)*v(302)*area+isc_coef*(area)*(temp-25)}
d1 301 300 diode
.model diode d(is={j0*area},eg=1.17)
d2 301 300 diode2
.model diode2 d(is={j02*area}, n=2)
rs 301 303 {rs}
rsh 301 300 {rsh}
.ends cell_4
```

(b) Calculate the response of a solar cell to a temperature of 80 °C, a temperature coefficient of the short-circuit current of 6.4×10^{-6} A/cm^2 °C and the same other parameters as in Example 3.4.

Solution

A '*.cir*' file has to be written:

```
*temp_2.cir
.include cell_4.lib
xcell3 0 31 32 cell_4 params:area=126.6 j0=1e-11 j02=0
+ jsc=0.0343 rs=1e-6 rsh=1000 temp=80 isc_coef=6.4e-6
vbias 31 0 dc 0
virrad 32 0 dc 1000
.plot dc i(vbias)
.dc vbias 0 0.6 0.01
.temp 80

.probe
.end
```



Figure 3.12 Solution to Example 3.5

The results for a temperature of 80 °C can be seen in Figure 3.12. A short circuit current of 4.38 A and an open-circuit voltage of 446 mV result from the PSpice simulation, according to the temperature coefficients used.

3.13 Effects of Space Radiation

The main effect of space radiation concerns the minority carrier lifetime degradation in the semiconductor bulk leading to an increase of the dark current density, and also a degradation of the photocurrent generated. Altogether, these effects produce a significant reduction of the maximum power of the solar cell, which is described analytically by empirical equations relating the values of the short-circuit current, open-circuit voltage and the maximum power to the fluence of a given high energy particle Φ. The units of the particle fluence are particles per unit surface and this magnitude has to be clearly differentiated from the photon flux ϕ used in Chapter 2,

$$J_{sc}(\Phi) = J_{sc}(BOL) - K_J \log\left(1 + \frac{\Phi}{\Phi_J}\right) \tag{3.53}$$

$$V_{oc}(\Phi) = V_{oc}(BOL) - K_V \log\left(1 + \frac{\Phi}{\Phi_V}\right) \tag{3.54}$$

$$P_{max}(\Phi) = P_{max}(BOL) - K_p \log\left(1 + \frac{\Phi}{\Phi_p}\right) \tag{3.55}$$

The values of the constants appearing in equations (3.53), (3.54) and (3.55) can be derived from the available data of solar cell degradation in space, which are known for a large variety of solar cells and particle type and fluences [3.3]. As an example, Table 3.7 shows the values of the constants for a silicon solar cell and a GaAs/Ge solar cell calculated from the

Table 3.7 Values of the space degradation constants for J_{sc}, V_{oc} and P_{max}

	Silicon solar cell 2 Ω cm BSFR			GaAs/Ge solar cell		
	Short circuit (mA/cm²)	Open circuit (mV)	Max. power (mW/cm²)	Short circuit (mA/cm²)	Open circuit (V)	Max. power (mW/cm²)
K_i	5.26	42	2.91	10.9	93.5	8.79
Φ_i(cm^{-2})	3.02×10^{13}	2.99×10^{12}	5.29×10^{12}	2.51×10^{14}	1×10^{14}	1.55×10^{14}

degradation graphs in reference [3.3]. K_i and Φ_i stand for K_I for the short-circuit current, K_V for the open-circuit voltage and K_p for the maximum power.

PSpice also allows the simulation of the $I(V)$ characteristics of a solar cell after a given radiation fluence. The inputs are the known values of J_{sc}, V_{oc} and P_{max} at the beginning of life (BOL) and the output is the full $I(V)$ characteristic resulting after a given space radiation of fluence Φ. The procedure to carry out this exercise consists of several steps:

Step nr 1: Calculate the value of J_0 in BOL conditions.

This can be accomplished recalling that in open circuit in the simplest solar cell model,

$$J = 0 = J_{sc} - J_0\left(e^{\frac{V_{oc}}{V_T}} - 1\right) \tag{3.56}$$

and then, aproximately, at BOL conditions

$$J_0(BOL) = J_{sc}(BOL)e^{-\frac{V_{oc}(BOL)}{V_T}} \tag{3.57}$$

Step nr 2: Calculate the ideal value of the FF in BOL.

This means the FF that the solar cell with zero series resistance would have. This is given by equation (3.17) reproduced here,

$$FF_0 = \frac{v_{oc} - \ln(v_{oc} + 0.72)}{1 + v_{oc}} \tag{3.58}$$

where

$$v_{oc} = \frac{V_{oc}}{V_T} \tag{3.59}$$

is the normalized value of the open-circuit voltage.

Step nr 3: Calculate the series resistance in BOL conditions.

This is easily accomplished by

$$FF(BOL) = FF_o(BOL)(1 - r_s) \tag{3.60}$$

after denormalizing r_s,

$$R_s = \frac{V_{oc}}{J_{sc} \times Area} - \frac{P_{max}}{FF_o(BOL) \times J_{sc}^2 \times Area} \tag{3.61}$$

Step nr 4: Write the solar cell model taking into account the degradation characteristics given by the constants in Table 3.6.

$$J_0(\Phi) = J_{sc}(\Phi)e^{-\frac{V_{oc}(BOL) - K_v \log\left(1+\frac{\Phi}{\Phi_v}\right)}{V_T}} \tag{3.62}$$

The short-circuit current can also be calculated using equation (3.53). It is assumed that the series resistance does not change after irradiation.

Example 3.6

Consider an $8\,\text{cm}^2$ silicon solar cell having been irradiated in space by a set of radiation fluences of 1×10^{10}, 1×10^{11}, 1×10^{12}, 1×10^{13}, 1×10^{14}, 1×10^{15}, 1×10^{16} 1MeV electrons/cm^2. If the BOL data are the following: $V_{oc} = 0.608$ V, $J_{sc} = 0.0436$ A/cm^2 and $P_{max} = 20.8$ mW/cm^2 simulate the PSpice $I(V)$ characteristics of the cell. Consider that the solar cell can be modelled by a single diode and has a shunt resistance of infinite value.

Before solving the problem with PSpice, we will illustrate here the steps previously described, in order to calculate some important magnitudes considering the values of the degradation constants in Table 3.7.

Step nr 1

From equation (3.57),

$$J_0(BOL) = J_{sc}(BOL)e^{-\frac{V_{oc}(BOL)}{V_T}} = 43.6 \times 10^3 e^{-\frac{0.608}{0.026}} = 17 \times 10^{-12}\,A$$

Step nr 2

$$v_{oc} = \frac{V_{oc}}{V_T} = \frac{0.608}{0.26} = 23.38$$

$$FF_0 = \frac{v_{oc} - \ln(v_{oc} + 0.72)}{1 + v_{oc}} = 0.828$$

Step nr 3

$$R_s = \frac{V_{oc}}{J_{sc} \times Area} - \frac{P_{max}}{FF_o(BOL) \times J_{sc}^2 \times Area} = \frac{0.608}{0.0436 \times 8} - \frac{0.02}{0.828 \times (0.0436)^2 \times 8} = 0.115\,\Omega$$

Step nr 4

For every fluence value, J_0 is calculated

$$J_0(\phi) = J_{sc}(\phi)e^{\frac{V_{oc}(BOL) - K_v \log\left(\frac{1+\phi}{\phi_v}\right)}{V_T}}$$

A PSpice subcircuit 'cell_5.lib' shown in Annex 3 is written, where it can be seen that equation (3.62) has been used to define the value of the 'is' parameter of the diode model. Moreover, the series resistance is internally calculated from equation (3.61).

The .cir file to calculate the full $I(V)$ characteristics is also shown in Annex 3 and is called 'space.cir'.

The correspondence between the names in the PSpice code and the names in the equations is the following:

Parameter in PSpice code	Parameter in equations (3.45) to (3.51)
vocbol	$V_{oc}(BOL)$
jscbol	$J_{sc}(BOL)$
fi	Φ_i
f	Φ
fv	Φ_v
kv	K_v
ki	K_i
jo	J
vocnorm	v_{oc}
uvet	V_T

The resulting $I(V)$ characteristics are shown in Figure 3.13.

The maximum power delivered by the solar cell can now be computed as 79.665 mW, that is 9.95 mW/cm^2. This value is approximately equal to the expected value of the maximum power after degradation as can be calculated using the degradation constants for the maximum power given in Table 3.7.



ıre 3.13 Degradation of the characteristics of the silicon solar cell of Example 3.6 for space iation fluences of 1×10^{10}, 1×10^{11}, 1×10^{12}, 1×10^{13}, 1×10^{14}, 1×10^{15}, 1×10^{16} electrons/cm^2

14 Behavioural Solar Cell Model

r the purpose of modelling photovoltaic systems, what is usually required are simulations the solar cell behaviour for changing temperature and irradiance conditions. This means ıt the built-in temperature analysis of PSpice is difficult to use because PSpice runs a new ılysis for every value of the temperature. Usually what is available as a result is a file or a ıe series of irradiance and temperature values. Moreover, if solar cells are not made of one the well-known semiconductors such as silicon, GaAs or Ge, the important physical data ȷuired to model the temperature effects of a solar cell are not easily available. Instead, ɔst frequently the data available come from data sheets or published material by the ınufacturers and concern values of electrical magnitudes from photovoltaic measurements, :h as the short-circuit current, open-circuit voltage and maximum power under some ndard conditions, and the temperature coefficients. If this is the case, the model of a solar ıl including a diode is not very practical and is of limited use. For this reason we introduce ɔehavioural model, based on voltage and current sources, which is able to correctly model : behaviour of an arbitrary solar cell under arbitrary conditions of irradiance and nperature, with electrical data values as the only input.

The model assumes that the solar cell can be modelled by two current sources and a series ;istance. The model is composed of a subcircuit between nodes 10 14 12 13 as shown in ȝure 3.14, where two g-devices are assigned two functions as follows:

$$i(girrad) = \frac{J_{scr}A}{1000}G + \left(\frac{dJ_{sc}}{dT}\right)(T_{cell} - T_r) \tag{3.63}$$

ıich returns the value of the short-circuit current at irradiance G. The derivative in equation 63) is the temperature coefficient of the short-circuit current and is considered constant in ɜ range of temperatures of interest. T_r is the reference temperature which is usually nsidered 25 °C (in some cases 300 °K are considered). This g-device is written as:

```
girrad 10 11 value={(jscr/1000*v(12)*area) +coef_jsc*area*(v(17)–25)}
```

The second source returns the exponential term of a solar cell replacing the diode

$$i(gidiode) = \frac{Jsc(area)}{\left(e^{\frac{V_{oc}}{V_T}} - 1\right)}\left(e^{\frac{V}{V_T}} - 1\right) \tag{3.64}$$

where the value of J_{sc} is copied from the voltage at node 305 and the open-circuit voltage is copied from node 306.

The value of the temperature involved in equation (3.64) has to be the cell operating temperature T_{cell}, which is usually derived from the NOCT concept. NOCT stands for nominal operating conditions temperature and is the temparature of the cell at 800 W/m^2 irradiance and 20 °C of ambient temperature.

$$T_{cell} - T_a = \frac{NOCT - 20}{800}G \tag{3.65}$$

This is written in PSpice as can be seen in Figure 3.14.

The open-circuit voltage can also be written for the new temperature and irradiance values.

The temperature coefficient of the open-circuit voltage is computed for the same value of the irradiance at two temperatures,

$$\left(\frac{\partial V_{oc}}{\partial T}\right)_G (T_{cell} - T_r) = V_T \ln\left(1 + \frac{I_{scr}}{I_0}\right) - V_{ocr} \tag{3.66}$$

Adding and subtracting the value of the V_{oc} under arbitrary radiance and temperature conditions

$$\left(\frac{\partial V_{oc}}{\partial T}\right)_G (T_{cell} - T_r) = V_T \ln\left(1 + \frac{I_{scr}}{I_0}\right) - V_{ocr} + V_{oc} - V_{oc} \tag{3.67}$$

Substituting

$$V_{oc} = V_T \ln\left(1 + \frac{I_{sc}}{I_0}\right) \tag{3.68}$$

it follows,

$$V_{oc} = V_{ocr} + \left(\frac{\partial V_{oc}}{\partial T}\right)_G (T_{cell} - T_r) + V_T \ln\frac{1 + \frac{I_{sc}}{I_0}}{1 + \frac{I_{scr}}{I_0}} \tag{3.69}$$



Figure 3.14 Schematic and block diagram of the solar cell behavioural model

In the general case, except when the short-circuit current is zero, unity can be neglected in e numerator and denominator of the last term in equation (3.69) and then

$$V_{oc} \approx V_{ocr} + \left(\frac{\partial V_{oc}}{\partial T}\right)_G (T_{cell} - T_r) + V_T \ln \frac{I_{sc}}{I_{scr}} \tag{3.70}$$

here V_T is calculated at the cell temperature T_{cell}.
The series resistance is calculated,

$$R_s = \frac{V_{oc}}{I_{sc}} - \frac{P_{\max}}{FF_0 I_{sc}^2} \tag{3.71}$$

here FF_o is given by equation (3.17), and P_{max} is the maximum power per unit area. The Spice code is written as,

```
rs n+ n- {vocr/(jscr*area)-pmaxr/(jscr**2*area*(vocr/0.026-log((vocr/0.026)
  +0.72))/(1+vocr/0.026))}
```


where the value is calculated at the reference temperature conditions and assumed independent of the cell temperature.

In PV systems the information about the evolution of the coordinates of the maximum power point during a given period of time is important because the electronic equipment used to connect the photovoltaic devices to loads are designed to follow this maximum power point. The PSpice code can also provide this information in two accessible nodes of the cell subcircuit. All we need are the coordinates of the maximum power point at standard conditions V_{mr} and I_{mr}. This is available from most of the data sheets provided by the manufacturers. In that case the current of the maximum power point at arbitrary conditions of irradiance and temperature can be considered to scale proportionally with the irradiance and linearly with the temperature with the temperature coefficient of the short-circuit current:

$$I_m = I_{mr}\frac{G}{G_r} + (area)Coef\_j_{sc}(T_{cell} - T_r) \tag{3.72}$$

and from this value the voltage of the maximum power point can be calculated taking into account that

$$I_m = I_{sc} - I_o\left(e^{\frac{V_m + I_m R_s}{V_T}} - 1\right) \tag{3.73}$$

and then

$$V_m = V_T \ln\left(1 + \frac{I_{sc} - I_m}{I_{sc}}\left(e^{\frac{V_{oc}}{V_T}} - 1\right)\right) - I_m R_s \tag{3.74}$$

The equations described above can be implemented in a PSpice library file 'cell_beh-lib' listed in Annex 3 and schematically shown in Figure 3.14.

Example 3.7

Consider a CIGS (copper indium gallium diselenide) solar cell with the following electrical characteristics: $V_{ocr} = 0.669$, $J_{scr} = 35.7\ \text{mA/cm}^2$, $P_{maxr} = 18.39\ \text{mW/cm}^2$. These values have been taken from M.A. Green, K. Emery, D.L. King, S. Igari, W. Warta 'Solar cell efficiency tables (version 18)' *Progress in Photovoltaics*, pp. 287–193, July-August 2001.

The temperature coefficients are for the short-circuit current +12.5 μA/cm² °C and for the open-circuit voltage −3.1 mV/°C. These values have been estimated for a 1 cm² cell from www.siemenssolar.com/st5.html.

The PSpice model 'cell_beh.lib 'is used in the file 'cell_beh.cir' also listed in Annex 3 to plot the $I(V)$ characteristics. The result is shown in Figure 3.15, where it can be seen that the model is able to simulate the $I(V)$ characteristic.



Figure 3.15 CIGS solar cell $I(V)$ characteristic simulated using the behavioural solar cell model

3.15 Use of the Behavioural Model and PWL Sources to Simulate the Response to a Time Series of Irradiance and Temperature

The behavioural model described in the previous section is aimed at simulating the response of a given solar cell to a certain profile of combined irradiance–ambient temperature time series. Although in the next chapters we will be extensively using time series of these variables it is convenient here to address two important points that help understanding of the PSpice simulation of PV systems, and not only solar cells: (a) time units and (b) variable units.

3.15.1 Time units

The operation of PV systems is meant to last over long periods of time, and the design and sizing procedures will be much assisted if long periods of time could be simulated considering arbitrary irradiance and temperature profiles. Apparently, such simulations require long CPU time, are unavailate or cumbersome. An easy way to overcome these problems is to consider two different units of time: one unit of time will be the *internal PSpice unit of time* and the second the *real PV system operating time*. This means that if we assign 1 microsecond of internal PSpice time to 1 hour of real time, a one-day simulation will be performed in a few seconds on any standard PC computer. This approach has been used in this book when necessary. In order to facilitate the understanding a warning has been placed in all figures with differences between PSpice internal time units and real time units.

3.15.2 Variable units

The use of input and output variables in the PSpice files used to simulate PV systems is required in order, for instance, to enter the values of irradiance and temperature, or to extract

form the simulation results the values of variables such as the state of charge of a battery or the angular speed of a motor. PSpice is, however, a tool that only handles electrical magnitudes. This means that the only internal variable units in PSpice are restricted to electrical units. In PV systems it is obvious that non-electrical magnitudes have to be handled and this then forces the use of electrical equivalents for non-electrical magnitudes. For example, if we want to enter temperature data we have to convert the temperature to a voltage source, for example, in such a way that 1 V of the voltage source corresponds to one degree Celsius of temperature. So the internal PSpice unit is an electrical unit (volt) and the real unit is a thermal unit (°C) therefore making a domain conversion. In order to avoid confusions, warnings are issued through the book when necessary. In fact we have used such domain conversion in Chapters 1 and 2 because we have handled spectral irradiance and wavelength, but we want to stress this domain conversion issue again here due to the importance it will take from now on.

As an example of the points raised above, let us consider an example assuming we know 10 irradiance and temperature couples of values for one day and we want to know the time evolution of the main PV magnitudes of a solar cell. The values of irradiance and temperature are entered in the PSpice code as a piecewise linear voltage source where the internal unit of time is considered as one microsecond corresponding to one hour of real time.

Example 3.8

Assume the following values of temperature and irradiance

Time	Irradiance (W/m^2)	Temperature (°C)
6	0	12
8	100	12
10	200	15
11	400	18
12	450	20
13	800	22
15	750	21
17	300	17
19	200	15
20	0	13

(a) Write the PWL sources and the PSpice file to compute the time evolution of the characteristics.

The subcircuit is changed modifying the v-sources for the irradiance and temperature as follows:

```
virrad node+ node- pwl 6u 0 8u 100 10u 200 11u 400 12u 450 13u 800 15u 750 17u 300 19u 200 20u 0
vtemp node+ node- pwl 6u 12 8u 12 10u 15 11u 18 12u 20 13u 22 15u 21 17u 17 19u 15 20u 13
```



Figure 3.16 Behavioural model and PWL sources for time series simulations. Warning: The internal unit of time is the microsecond and the real time unit is the hour. The internal units of the irradiance are volts and the real units are W/m^2, the internal units of temperature are volts and the real units are °C

(b) Write a PSpice .cir file calling a behavioural solar cell model and performing a transient analysis covering the 10 hours of the example.

```
*cell_pwl.cir

.include cell_beh.lib
xcellbeh 0 32 33 34 35 36 37 38 39 cell_beh params:area=1, tr=25, jscr=0.0375,
+ pmaxr=0.0184, vocr=0.669, jmr=35.52e-3, vmr=0.518, noct=47,
+ coef_jsc=12.5e-6, coef_voc=-3.1e-3

virrad 32 0 pwl 6u 1 8u 100 10u 200 11u 400 12u 450 13u 800 15u 750 17u 300 19u 200 20u 0
vtemp 33 0 pwl 6u 12 8u 12 10u 15 11u 18 12u 20 13u 22 15u 21 17u 17 19u 15 20u 13
rim 38 0 1
vbias 34 0 dc 0
.tran 0 20u 6u 0.01u
.probe
.end
```

The short circuit is included by means of an auxiliary voltage source vaux = 0 in order to be able to measure the short-circuit current. The result is shown in Figure 3.16, where it can be seen that the time evolution of irradiance and temperature can be correctly handled.

3.16 Problems

3.1 Using the single diode model of the solar cell with $r_s = 0$ and R_{sh} = infinity, draw the ideal $I(V)$ characteristic of a solar cell with $J_{sc} = 0.035$ for $I_0 = 1 \times 10^{-11}$, $I_0 = 1 \times 10^{-10}$ and $I_0 = 1 \times 10^{-9}$ at irradiance 1000 W/m^2. Find the values of the open-circuit voltage and short-circuit current.

3.2 One of the methods used to measure the series resistance of a solar cell is based on the comparison between the dark $I(V)$ curve and the couple of values I_{sc}–V_{oc} at several irradiance values. Write the PSpice code of a solar cell working at 27 °C with the following parameter values: $J_{sc} = 0.0343$, $j_{01} = 1 \times 10^{-11}$, $j_{02} = 0$, $r_{sh} = 1 \times 10^6\ \Omega$, $area = 126.6\ \text{cm}^2$ and $r_s = 0.05\ \Omega$. Plot the $I(V)$ characteristics for the following irradiance values: 200, 400, 600, 800, 1000, 1200 W/m^2. List the values for I_{sc} and V_{oc} for all irradiance values. Write the PSpice code of the same solar cell at the same temperature but in darkness. Plot the $I(V)$ characteristic in a semilog plot(log I vs. V). (Notice that the curve is not linear at high currents.) List the values of V in this plot for approximately the same I_{sc} values of the previous list. Verify that $(V - V_{oc})/I_{sc}$ gives the value of the series resistance.

3.3 Another method to measure the series resistance is based on two measures of the $I(V)$ characteristics of the solar cell at two irradiance values and at the same temperature. The purpose of this problem is to demonstrate this method.

First write the PSpice code of a solar cell with the parameters $j_{sc} = 0.0343$, $j_{01} = 1 \times 10^{-11}$, $j_{02} = 0$, $r_{sh} = 1 \times 10^6\ \Omega$, $area = 126.6\ \text{cm}^2$ and $r_s = 0.05\ \Omega$. Plot the $I(V)$ characteristic for two irradiance values: 600 and 1000 W/m^2. Using the cursor find a point on the curve drawn for 1000 W/m^2 where $I = I_{sc_1} - 1$ A and find the value of the voltage at this point. Call this value V_1.

Do the same with the 600 W/m^2 plot and find the value of the voltage V_2 at the point where $I = I_{sc_2} - 1$ A. Verify that $(V_2 - V_1)/(I_{sc_1} - I_{sc_2})$ gives the value of R_s.

3.17 References

[3.1] Van Overstraeten, R.J. and Mertens, R.P. *Physics Technology and Use of Photovoltaics*, Adam Hilgher, 1986.
[3.2] Green, M.A., *Solar Cells*, Bridge Printery, Rosebery, NSW, Australia, 1992.
[3.3] Anspaugh, B.E., *Solar Cell Radiation Handbook, Addendum 1*, NASA Jet Propulsion Laboratory publication JPL 82-69, 15 February, Pasadena, California, 1989, Figures 1, 2, 3, 21, 22 and 23.